Clear and Concise Code

Easy to Write and Simple to Change

```
* Example 2
* Invoice details input and validation
*
*PFILE  'Report File Output'
*DICT          CUSTOMERS  "C"  I CUS
* *IFILE  CUS X317                CUSTFL.SLD         MODE1
* *DICT
*   C0                :C317, etc.
*DICT          CONTROL  "AD"  IO CTL
* *IOFILE CTL V256,200,O  ACCS.DET
* *DICT
*   AD0               :C256, etc.
*DICT          INVOICE  ""    O INV
* *OFILE INV QF51          INVOICE.DET
* *DICT
*   INVOICE0          :C51
*     CUSTOMER    INVOICE0:I8     , INVOICE.NO    :I2    , VALUE     :I8.2
*     VAT                 :I8.2   , DISCOUNT      :I8.2  , QTY       :I8
*     PROD.CODE           :C3     , DATE          /6 !D1 , FILLER    INVOICE0+51:C0
*DICTIONARY
 REC.COUNT    %I2    , KEY       %I2    , WK4      %I8.2
 WK1          %I8.2 , WK2       %I8.2 , WK3      %I8.2
*LIST DETAILS
 A CUSTOMER,     B PROD.CODE,    C INVOICE.NO, 1 VALUE
 2 VAT,          3 DISCOUNT,     4 QTY,        5 REC.COUNT
 6 WK1,          7 WK2,          8 WK3,        9 WK4
*PICTURE HEAD CH1,1 '
Transactions Input
Customer Invoice No.        Product              Quantity         Nett        Vat  Discount       Gross
                                                                  Value                           Value

*PICTURE DATA 1,0
AAAAAA    CCCCC                BBB                  44444      £1111.11     £222.22   £333.33     £9999.99
*PICTURE LAST 2,0 '
Total for Run'                                                 £66666.66    £7777.77  £8888.88    £99999.99    '
Number of records input'        555    '
Next invoice number'          CCCCC

*DETAB INITIAL
I PRINT          HEAD     DETAILS              [Print heading
  REC.COUNT      Z                             [Zeroise record count
  DATE           MV_DAT   'TODAY'              [Set up todays date
  KEY            Z
  LOOKUP         CTL      KEY                  [Read first element of
  READ           CTL      AD0                  [ ACCS.DET file
  INVOICE.NO     MV       ADINV.NO             [PICK UP INVOICE NO.
  GOTO SECOND
*
*DETAB SECOND
I INPUTP         CUSTOMER,PROD.CODE                      [Inputs & checks lengths
C CUSTOMER       =        0                  N N N Y     [End Run?
  LOOKUP         CUS      CUSTOMER           N - Y -     [Is customer on file?
  PROD.CODE      V        '00N'              - N Y -     [Is PROD.CODE valid?
A DISPLAY        'CUSTOMER NOT ON FILE'      X - - -     [
  DISPLAY        'PRODUCT INVALID'           - X - -     [
  INPUTP         QTY,VALUE,VAT,DISCOUNT      - - X -     [Inputs & checks fields
  WK4            MV       VALUE              - - X -     [Calculate
  WK4            +        VAT                - - X -     [ Gross
  WK4            -        DISCOUNT           - - X -     [ Value
  PRINT          DATA DETAILS                - - X -     [
  WRITE          INV      INVOICE0           - - X -     [Write Record to INVOICE.DET
  REC.COUNT      +        1                  - - X -     [Increment record count
  INVOICE.NO     +        1                  - - X -     [
  WK1            +        VALUE              - - X -     [
  WK2            +        VAT                - - X -     [
  WK3            +        DISCOUNT           - - X -     [
  GOTO           SECOND                      X X X -     [Loop to input more
  WK4            MV       WK1                - - - X     [Calculate
  WK4            +        WK2                - - - X     [ Overall
  WK4            -        WK3                - - - X     [ Gross Value
  PRINT                   LAST DETAILS        - - - X     [Print final line
  ADINV.NO       MV       INVOICE.NO         - - - X     [Store next INVOICE.NO
  REWRITE        CTL      AD0                - - - X     [ in ACCS.DET
  DELETE                  'XX'               - - - X     [Terminate Run
*GO
```

EXAMPLE OF USER DIALOGUE

```
Report File Output? JIRPL5.OUT
CUSTOMER,PROD.CODE
? 219768,112
QTY,VALUE,VAT,DISCOUNT
? 20,50.00,4.0,0
CUSTOMER,PROD.CODE
? 219768,124
QTY,VALUE,VAT,DISCOUNT
? 300,75.00,6.00,0
CUSTOMER,PROD.CODE
? 219788,E19
CUSTOMER NOT ON FILE
PRODUCT INVALID
CUSTOMER,PROD.CODE
? 219768,019
QTY,VALUE,VAT,DISCOUNT
? 3DOZ,57.00,8.00,0
QTY INVALID. RE-INPUT
? 36
Etc.
```

```
Transactions Input
Customer Invoice No.        Product              Quantity         Nett        Vat  Discount       Gross
                                                                  Value                           Value
219768      2192                112                    20      £50.00      £4.00     £0.00      £54.00
219768      2193                124                   300      £75.00      £6.00     £0.00      £81.00
219768      2194                019                    36      £57.00      £8.00     £0.00      £65.00
211313      2195                143                    36      £60.00      £4.00     £5.00      £59.00
211313      2196                156                   500      £80.00      £5.10     £0.00      £65.10
211313      2197                161                    25      £49.50      £3.00     £0.00      £34.50

Total for Run                                                  £371.50     £32.10    £5.00      £398.60
Number of records input           6
Next invoice number            2198
```

*PFILE

Defines the file to which the output report is directed. No file has been defined in this case, but the user is to be prompted at run time with "Report File output" to input the file or device required e.g. KB: or LP0: or INVS.OUT.

*DICTIONARY OR *DICT

Can be used in 2 ways: With parameters, it takes predefined information from the data dictionary and automatically generates file and field definitions. In the example these are shown as commented statements.

Alternatively the programmer can associate user defined names with fields on files, or fields in the work area which are to be used for temporary storage. An integer field REC.COUNT is used to maintain a count of the number of valid records input. KEY is used to access the ACCS.DET virtual array file. WK1, WK2 and WK3 are used as accumulation areas for Nett Value, VAT and Discount respectively and WK4 is used to compute the Gross Value.

*IFILE

Defines the customer file as an input indexed sequential file. This is commented out as part of the *DICT statement.

*IOFILE

Defines a virtual array file to be used as an input and output file, for the next free invoice number. This is commented out as part of the *DICT statement.

*OFILE

Defines the sort compatible output file containing the invoice details. This is commented out as part of the *DICT statement.

*LIST

Lists fields to be printed by associating a single character (known as a field specifying character) from the available character set with fields defined in *DICTIONARY

*PICTURE

Defines the format of headings and data to be printed. Here *PICTURE HEAD defines the heading to be printed, the operands CHI,1 define that the heading is to be printed by throwing to channel 1 (form feed) and feeding 1 line after the heading. PICTURE DATA defines the format of the data to be printed using field specifying characters defined in *LIST. *PIC LAST will be used to print final lines of report.

*DETAB INITIAL

Is a decision table entered once at the start of the run. It prints the report heading, sets up the date as current date and reads the first element of virtual array file ACCS.DET.

*DETAB SECOND

Contains the main processing loop for the program. Initially it prompts the user to input Customer number and Product Code. The Customer Number is looked up on the Indexed sequential file CUSTFL.SLD to check that it is present. Product Code is validated using the validate operator V against a string of pattern characters that check that the first 2 characters are numeric (pattern character '0') and the third character is alphanumeric (pattern character 'N'). (Various pattern characters are available in RPL to simplify validation procedures.) If Customer Number or Product Code are invalid the user is prompted to re-input them (rules 1 or 2).

If Customer Number and Product Code are valid the user is prompted to input the remaining fields relevant to the invoice. Note that the INPUTP verb which requests input and prompts the user at the keyboard also validates the lengths of fields input and checks that integer and floating point fields are valid. The INPUTP verb allows multiple input fields and prompts only for specific fields failing the automatic validation check—fields can also be omitted on input and RPL space or zero fills as appropriate. Details of the invoice are listed on the report file and written to the file INVOICE.DET.

When the user inputs a Customer number of zero, to indicate end of run, rule 4 is entered and the total Gross Value is computed, the final output line is printed, the next invoice number to be used is stored in ACCS.DET file and the program terminated by the DELETE verb.

*GO Terminates the parameters and causes the program to be run.